
写在前面的话

本书起源于中国电子工业出版社出版的一本书：《自己动手写操作系统》（于渊著）。我对《自己

动手写操作系统》这本书中使用商业软件做为演示平台比较惊讶，因为不是每个人都买得起正版软件

的，尤其是穷学生。我想《自》所面向的主要受众也应该是学生，那么一本介绍只有商业软件才能实现

的编程技巧的书将会逼着穷学生去使用盗版，这是非常罪恶的行为 /。
由于本人是一个 Linux用户，一个开源软件的拥护者，所以就试着使用开源软件实现这本书中的所

有 demo，并在自己的博客上进行推广。后来我觉得，为什么我不能自己写本书呢？这样我就能插入漂
亮的插图，写更详尽的介绍而不怕篇幅过长，更容易让读者接受也更容易传播，所以我就开始写这本《

使用开源软件自己动手写操作系统》。

定下写一本书的目标毕竟不像写一篇博客，我将尽量详尽的介绍我使用的方法和过程，以图能让不

同技术背景的读者都能通畅地完成阅读。但是自己写并且排版一本书不是很轻松的事情，需要耗费大量

时间，所以我只能抽空一点一点的将这本书堆砌起来，这也是您之所以在本页上方看到本书版本号的原

因 ,。
本书的最终目标是成为一本大学“计算机操作系统”课程的参考工具书，为学生提供一个 step by

step的引导去实现一个操作系统。这不是一个容易实现的目标，因为我本人现在并不自信有那个实力了
解操作系统的方方面面。但是我想，立志百里行九十总好过于踯躅不前。

《自己动手写操作系统》一书开了个好头，所以在前面部分，我将主要讨论使用开源软件实现

《自》的 demo。如果您有《自》这本书，参考阅读效果会更好，不过我将尽我所能在本书中给出清楚
的讲解，尽量使您免于去参考《自》一书。

出于开放性和易编辑性考虑，本书采用 LATEX排版，在成书前期由于专注于版面，代码比较杂乱，
可读性不强，暂不开放本书 TEX源代码下载。

如果您在阅读过程中有什么问题，发现书中的错误，或者好的建议，欢迎您使用我留下的联系方式

与我联系，本人将非常感谢。

杨文博

电子邮件: solrex@gmail.com
个人主页：http://solrex.cn

个人博客：http://blog.solrex.cn

2008年 1月 9日

序言

这里应该是各个章节的摘要和版式简介，不过因为写摘要向来是件让人心烦的事情，所以我准备把

它放在最后写 ,。

CHAPTER 1

计算机启动

任何一个计算机软件都是由一系列的可执行文件组成的，可执行文件的内容是可以被机器识别的二

进制指令和数据。一般可执行文件的运行是在操作系统的照看下加载进内存并运行的，由操作系统给它

分配资源和处理器时间，并确定它的执行方式。操作系统也是由可执行文件组成的，但是操作系统的启

动方式和一般应用软件是不同的，这也就是它叫做“操作系统”的原因 ,。
没有操作系统的机器，一般情况下被我们称为“裸机”，意思就是只有硬件，什么都干不了。但是

一个机器怎么知道自己是不是裸机呢？它总要有方式去判断机器上安装没有安装操作系统吧。下面我们

就简单介绍一下计算机启动的过程：

1.1 计算机启动过程

计算机启动过程一般是指计算机从点亮到加载操作系统的一个过程。对于 IBM兼容机（个人电
脑）来讲，这个过程大致是这样的：

1. 加电 电源开关被按下时，机器就开始供电，主板的控制芯片组会向 CPU（Central Processing
Unit，中央处理器）发出并保持一个 RESET（重置）信号，让 CPU恢复到初始状态。当芯片组
检测到电源已经开始稳定供电时就会撤去 RESET信号（松开台式机的重启键是一样的效果），
这时CPU就从 0xffff0处开始执行指令。这个地址在系统BIOS（Basic Input/Output System，
基本输入输出系统）的地址范围内，大部分系统 BIOS厂商放在这里的都只是一条跳转指令，跳
到系统 BIOS真正的启动代码处。

2. 自检 系统 BIOS 的启动代码首先要做的事情就是进行 POST （Power-On Self Test，加电后自
检），POST的主要任务是检测系统中一些关键设备是否存在和能否正常工作，例如内存和显卡
等。由于 POST是最早进行的检测过程，此时显卡还没有初始化，如果系统 BIOS在 POST的
过程中发现了一些致命错误，例如没有找到内存或者内存有问题（此时只会检查 640K 常规内
存），那么系统 BIOS就会直接控制喇叭发声来报告错误，声音的长短和次数代表了错误的类
型。

3. 初始化设备 接下来系统 BIOS将查找显卡的BIOS，存放显卡 BIOS的 ROM芯片的起始地址通
常设在 0xC0000处，系统 BIOS在这个地方找到显卡 BIOS之后就调用它的初始化代码，由显卡

1



2 1.2. 磁盘抽象物理结构

BIOS来初始化显卡，此时多数显卡都会在屏幕上显示出一些初始化信息，介绍生产厂商、图形
芯片类型等内容。系统 BIOS接着会查找其它设备的 BIOS程序，找到之后同样要调用这些 BIOS
内部的初始化代码来初始化相关的设备。

4. 测试设备 查找完所有其它设备的 BIOS之后，系统 BIOS将显示出它自己的启动画面，其中包括
有系统 BIOS的类型、序列号和版本号等内容。接着系统 BIOS将检测和显示 CPU的类型和工
作频率，然后开始测试所有的 RAM（Random Access Memory，随机访问存储器），并同时在
屏幕上显示内存测试的进度。内存测试通过之后，系统 BIOS将开始检测系统中安装的一些标准
硬件设备，包括硬盘、光驱、串口、并口、软驱等，另外绝大多数较新版本的系统 BIOS在这一
过程中还要自动检测和设置内存的定时参数、硬盘参数和访问模式等。标准设备检测完毕后，

系统 BIOS内部的支持即插即用的代码将开始检测和配置系统中安装的即插即用设备，每找到一
个设备之后，系统 BIOS都会在屏幕上显示出设备的名称和型号等信息，同时为该设备分配中断
（INT）、DMA（Direct Memory Access，直接存储器存取）通道和 I/O（Input/Output，输入
输出）端口等资源。

5. 更新 SECD 所有硬件都检测配置完毕后，多数系统 BIOS会重新清屏并在屏幕上方显示出一个
表格，其中概略地列出了系统中安装的各种标准硬件设备，以及它们使用的资源和一些相关工作

参数。接下来系统 BIOS将更新 ESCD（Extended System Configuration Data，扩展系统配置数
据）。 ESCD是系统 BIOS用来与操作系统交换硬件配置信息的一种手段，这些数据被存放在
CMOS（Complementary Metal Oxide Semiconductor，互补金属氧化物半导体）之中。

6. 启动操作系统 ESCD 更新完毕后，系统 BIOS 的启动代码将进行它的最后一项工作，即根据
用户指定的启动顺序从软盘、硬盘或光驱启动操作系统。以 Windows XP 为例，系统 BIOS
将启动盘（一般是主硬盘）的第一个扇区（Boot Sector，引导扇区）读入到内存的 0x7c00

处，并检查 0x7dfe地址的内存，如果其内容是 0xaa55，跳转到 0x7c00处执行 MBR（Master
Boot Record，主引导记录），MBR 接着从分区表（Partition Table）中找到第一个活动分区
（Active Partition，一般是 C盘分区），然后按照类似方式读取并执行这个活动分区的引导扇区
（Partition Boot Sector），而引导扇区将负责读取并执行 NTLDR（NT LoaDeR，Windows NT
的加载程序），然后主动权就移交给了Windows。

从以上介绍中我们可以看到，在第 6步之前，电脑的启动过程完全依仗于系统 BIOS，这个程序一
般是厂商写就固化在硬盘上的。我们所需要做的，就是第 6步之后的内容，即：

如何写一个操作系统并把它加载到内存?

1.2 磁盘抽象物理结构

由于操作系统的启动涉及到硬件地址写入和磁盘文件寻找，为了更好理解内存地址和文件存储的相

关知识，我们先来了解一下磁盘的结构。

1.2.1 硬盘

Figure 1.1: 硬盘

图 1.1所示就是硬盘（如非特指，我们这里的“硬盘”一般指代磁介质非固态硬盘）的外观图。其
中左边是硬盘盒拆开后盘片、磁头和内部机械结构的透视图，右边是普通台式机硬盘的外观图。现在的

硬盘容量较以前已经有大幅度增加，一般笔记本电脑硬盘容量已经在 120G以上，台式机硬盘容量一般
也达到了 160G大小。一般情况下，硬盘都是由坚硬金属材料（或者玻璃等）制成的涂以磁性介质的盘
片构成的，一般有层叠的多片，每个盘片都有两个面，两面都可以记录信息。

Figure 1.2: 硬盘的抽象物理结构

图 1.2为硬盘的抽象物理结构，需要注意的是这并不是硬盘真正的物理构造，所以这里我们称其为
“抽象”物理结构。因此我们下面讨论的也不是真正的硬盘技术实现，仅仅就硬盘（以及软盘等类似磁

介质存储器）存储结构以程序员易于理解的角度进行简单的介绍。

如图 1.2所示，硬盘是由很多盘片组成的，那些上下有分割的圆盘就表示一个个盘片。每个盘片被
分成许多扇形的区域，每个区域叫一个扇区，通常每个扇区存储 512字节（ FAT文件格式），盘片表
面上以盘片中心为圆心，不同半径的同心圆称为磁道。硬盘中，不同盘片相同半径的磁道所组成的圆柱

称为柱面。磁道与柱面都是表示不同半径的圆，在许多场合，磁道和柱面可以互换使用。每个磁盘有两

个面，每个面都有一个磁头，习惯用磁头号来区分。扇区，磁道（或柱面）和磁头数构成了硬盘结构的

基本参数，使用这些参数可以得到硬盘的容量，其计算公式为：

存储容量＝磁头数×磁道（柱面）数×每磁道扇区数×每扇区字节数

要点：

• 硬盘有数个盘片，每盘片两个面，每面一个磁头。

• 盘片被划分为多个扇形区域即扇区。

• 同一盘片不同半径的同心圆为磁道。

• 不同盘片相同半径构成的圆柱面即柱面。

• 公式：存储容量＝磁头数×磁道（柱面）数×每道扇区数×每扇区字节数。

• 信息记录可表示为：××磁道（柱面），××磁头，××扇区。

1.2.2 软盘

由于我们在本书中主要使用软盘作为系统启动盘，所以下面对应于硬盘介绍一下软盘的相关知识。

Figure 1.3: 软盘

现在通常能看到的软盘主要是 3.5英寸软盘，3.5英寸指的是其内部磁介质盘片的直径。从存储结
构上来讲，软盘与硬盘的主要不同就是软盘只有一个盘片且其存储密度较低。

由于软盘只有一个盘片，两个面，所以 3.5英寸软盘的容量可以根据上一小节的公式算出：

2(磁头)× 80(磁道)× 18(扇区)× 512 bytes(扇区的大小) = 2880 x 512 bytes = 1440 KB = 1.44MB

在这里需要引起我们特别注意的就是第 0号磁头（面），第 0号磁道的第 0号扇区，这里是一切的
开始。

1.2.3 启动扇区

软盘是没有所谓的 MBR的，因为软盘容量较小，没有所谓的分区，一张软盘就显示为一个逻辑
磁盘。当我们使用软盘启动电脑的时候，系统从软盘中首先读取的就是第一个扇区，即前面所说的第

0面，第 0号磁道的第 0号扇区，如果这个扇区的最后两个字节是 0xaa55，这里就简单叫做启动扇区

（Boot Sector）。所以我们首先要做的就是：在启动扇区的开始填入需要被执行的机器指令；在启动
扇区的最后两个字节中填入 0xaa55，这样这张软盘就成为了一张可启动盘。

� 启动扇区最后两个字节的内容为 0xaa55，这种说法是正确的――当且仅当表 2.1中
的 BPB BytesPerSec（每扇区字节数）的值为 512。如果 BPB BytesPerSec的值大于

512， 0xaa55的位置不会变化，但已经不是启动扇区最后两个字节了。

整个过程如图 1.4所示：

Figure 1.4: 启动扇区加载示意图

需要注意的是，软盘的启动扇区并不像一个文件一样，可以直接读取，写入启动扇区的过程是需要

一些技巧的，下面我们将讨论如何去实现。

1.3 使用虚拟机

在实现一个简单的操作系统时，我们是不可能拿一台真正的机器做实验的，一是很少有人有这个条

件，还有就是那样做比较麻烦。所以我们使用虚拟机来模拟一台真实的电脑，这样我们就能直接用虚拟

机加载软盘镜像来启动了，而制作软盘镜像显然要比写一张真实的软盘简单许多。

在 Linux 下有很多虚拟机软件，我们选择 VirtualBox 和 Bochs 作为主要的实现平台，我们用
VirtualBox做 demo，而 Bochs主要用作调试。下面给出一些虚拟机设置的指导，其实用哪种虚拟机都
没有关系，我们需要的只是虚拟机支持加载软盘镜像并能从软盘启动。

1.3.1 VirtualBox

VirtualBox 是遵从 GPL 协议的开源软件，它的官方网站是 http://www.virtualbox.org 。

VirtualBox 的官方网站上提供针对很多 Linux 系统平台的二进制安装包，比如针对 Red Hat 系列
（Fedora, RHEL）各种版本的 RPM安装包，针对 Debian系（Debian, Ubuntu）各种版本的 DEB安
装包，其中Ubuntu Linux可以更方便地从Ubuntu软件仓库中直接下载安装： sudo apt-get install

virtualbox。

安装好 VirtualBox后，需要使用 sudo adduser ‘whoami‘ vboxusers（某些系统中的添加用户命

令可能是 useradd）将自己添加到 VirtualBox的用户组 vboxusers中去；当然，也可以使用 GNOME
或者 KDE的图形界面用户和组的管理工具来添加组用户，也可以直接编辑 /etc/group文件，将自己的
用户名添加到 vboxusers对应行的最后，例如 vboxusers:x:501:solrex，部分 Linux可能需要注销后
重新登录当前用户。

我们下面使用 CentOS上安装的 VirtualBox演示如何用它建立一个虚拟机。

第一次启动 VirtualBox，会首先弹出一个 VirtualBox个人使用协议 PUEL的对话框（某些版本的
Linux可能不会弹出）：

Figure 1.5: VirtualBox个人使用协议

http://www.virtualbox.org

阅读完协议后，将下拉条拉到最低可以激活最下方的同意按钮，点击之：

Figure 1.6: 同意 VirtualBox个人使用协议

弹出的 VirtualBox用户注册对话框，可忽视关闭之：

Figure 1.7: VirtualBox用户注册对话框

接下来我们就见到了 VirtualBox主界面：

Figure 1.8: VirtualBox主界面

点击 New按钮新建一个虚拟机：

Figure 1.9: 新建一个虚拟机

我们使用 solrex作为虚拟机的名字，系统类型未知：

Figure 1.10: 设置虚拟机名字和操作系统类型

设置虚拟机的内存容量，这里随便设了 32M：

Figure 1.11: 设置虚拟机内存容量

设置虚拟机硬盘镜像：

Figure 1.12: 设置虚拟机硬盘镜像

如果没有硬盘镜像，需点“New”新建一块硬盘镜像：

Figure 1.13: 新建一块虚拟硬盘

点“Next”，设置虚拟硬盘镜像为可自动扩充大小：

Figure 1.14: 设置虚拟硬盘类型

这里将虚拟硬盘镜像的名字设置为“solrex”，并将容量设置为“32M”：

Figure 1.15: 设置虚拟硬盘镜像名字和容量

最后查看新建的虚拟硬盘信息，点击 Finish确认新建硬盘镜像：

Figure 1.16: 虚拟硬盘信息

令虚拟机使用已建立的虚拟硬盘 solrex.vdi：

Figure 1.17: 使用新建的虚拟硬盘

最后查看新建的虚拟机信息，点击 Finish确认新建虚拟机：

Figure 1.18: 虚拟机信息

回到 VirtualBox主界面，左侧列表中有新建立的虚拟机 solrex：

Figure 1.19: 回到 VirtualBox主界面



14 1.4. 使用软盘镜像

1.3.2 Bochs

1.4 使用软盘镜像

1.4.1 制作软盘镜像

前面我们说过，软盘的结构比较简单，所以我们选择使用软盘镜像来启动虚拟计算机。在 Linux下
制作一个软盘镜像很简单，只需要使用：

$ dd if=/dev/zero of=emptydisk.img bs=512 count=2880

命令就可以在当前目录下生成一个名为 emptydisk.img的空白软盘镜像，下面我们使用这个空白软盘

镜像来启动虚拟机。

dd：转换和拷贝文件的工具。 dd可以设置很多拷贝时候的参数，在本例中 if=FILE选项
代表从 FILE中读取内容； of=FILE选项代表将导出输出到 FILE； bs=BYTES代表每次读
取和输出 BYTES个字节； count=BLOCKS代表从输入文件中共读取 BLOCKS个输入块。
而这里的 /dev/zero则是一个 Linux的特殊文件，读取这个文件可以得到持续的 0。那么

上面命令的意思就是以每块 512字节共 2880块全空的字符填入文件 emptydisk.img中。我们
注意到前面提及的软盘容量计算公式：

2(磁头)× 80(磁道)× 18(扇区)× 512 bytes(扇区的大小) = 2880 x 512 bytes = 1440 KB
= 1.44MB
可以发现我们用上述命令得到的就是一张全空的未格式化的软盘镜像。

1.4.2 用软盘镜像启动虚拟机

在虚拟机主界面选中虚拟机后点 Settings按钮，进入虚拟机的设置界面：

Figure 1.20: 虚拟机设置界面

在左侧列表中选择 Floppy进入虚拟机软盘设置界面：

Figure 1.21: 虚拟机软盘设置

点击 Image File最右侧的文件夹标志，进入选择软盘镜像界面：

Figure 1.22: 选择软盘镜像

点击 Add按钮添加新的软盘镜像 emptydisk.img，并点击 select按钮选中其作为启动软盘：

Figure 1.23: 选择启动软盘镜像

返回虚拟机软盘设置界面后，点击 OK确认镜像文件信息：

Figure 1.24: 确认启动镜像软盘文件信息

返回虚拟机主界面，查看右侧的虚拟机设置信息：

Figure 1.25: 查看虚拟机设置信息

选中虚拟机后，双击或点击 Start按钮运行它，第一次运行可能给出如下信息：

Figure 1.26: 自动键盘捕获警告信息

这个对话框的意思就是，当鼠标在虚拟机内部点击时，鼠标和键盘的消息将被虚拟机自动捕获，成

为虚拟机的键盘和鼠标，可以敲击键盘右侧的 Ctrl键解除捕获。

显示虚拟机的运行时内容：

Figure 1.27: 虚拟机运行时

我们可以看到在图 1.27 中，虚拟机加载空白软盘启动后提示消息为：“FATAL: No bootable
medium found! System halted.”，换成中文是找不到可启动媒体，系统停机。它的实际意思就是在前
面第 1.1节“计算机启动过程”中提到的第 6步中虚拟机遍历了软驱、光驱、硬盘后没有找到可启动的
媒体，所以就只好停机。因为我们在启动前已经在软驱中加载了软盘镜像，所以提示信息就表明那个软

盘镜像不具有启动系统的功能，那么如何才能创建一个可启动的软盘呢，我们将在第 2章介绍。



CHAPTER 2

最小的“操作系统”

任何一个完善的操作系统都是从启动扇区开始的，这一章，我们就关注如何写一个启动扇区，以及

如何将其写入到软盘镜像中。

先介绍一下需要使用的工具：

• 系统： Cent OS 5.1(RHEL 5.1)

• 使用工具： gcc, binutils(as, ld, objcopy), dd, make, hexdump, vim, virtualbox

2.1 Hello OS world!

�
本章节内容需要和 gcc, make相关的 Linux C语言编程以及 PC汇编语言的基础知识。

很多编程书籍给出的第一个例子往往是在终端里输出一个字符串“Hello world!”，那么要写操作
系统的第一步给出的例子自然就是如何在屏幕上打印出一个字符串喽。所以，我们首先看《自己动手写

操作系统》一书中给出的第一个示例代码，在屏幕上打印“Hello OS world!”：

1 org 07c00h ; 告诉编译器程序加载到7c00处

2 mov ax, cs

3 mov ds, ax

4 mov es, ax

5 call DispStr ; 调用显示字符串例程

6 jmp $ ; 无限循环

7 DispStr:

8 mov ax, BootMessage

9 mov bp, ax ; ES:BP = 串地址

10 mov cx, 16 ; CX = 串长度

11 mov ax, 01301h ; AH = 13, AL = 01h

12 mov bx, 000ch ; 页号为0(BH = 0) 黑底红字(BL = 0Ch,高亮)

13 mov dl, 0

19



20 2.1. Hello OS world!

14 int 10h ; 10h 号中断

15 ret

16 BootMessage: db "Hello, OS world!"

17 times 510-($-$$) db 0 ; 填充剩下的空间，使生成的二进制代码恰好为512字节

18 dw 0xaa55 ; 结束标志

Figure 2.1: 《自》第一个实例代码 boot.asm

2.1.1 Intel汇编转化为 AT&T(GAS)汇编

上面 boot.asm中代码使用 Intel风格的汇编语言写成，本也可以在 Linux下使用同样开源的 NASM
编译，但是鉴于很少有人在 Linux下使用此汇编语法，它在 Linux平台上的扩展性和可调试性都不好（
GCC不兼容），而且不是采用 Linux平台上编译习惯，所以我把它改成了使用 GNU工具链去编译连
接。这样的话，对以后使用 GNU工具链编写其它体系结构的 bootloader也有帮助，毕竟 NASM没有
GAS用户多（也许 ,）。
上面的汇编源程序可以改写成 AT&T风格的汇编源代码：

1 .code16 #使用16位模式汇编

2 .text #代码段开始

3 mov %cs,%ax

4 mov %ax,%ds

5 mov %ax,%es

6 call DispStr #调用显示字符串例程

7 jmp . #无限循环

8 DispStr:

9 mov $BootMessage, %ax

10 mov %ax,%bp #ES:BP = 串地址

11 mov $16,%cx #CX = 串长度

12 mov $0x1301,%ax #AH = 13, AL = 01h

13 mov $0x00c,%bx #页号为0(BH = 0) 黑底红字(BL = 0Ch,高亮)

14 mov $0,%dl

15 int $0x10 #10h 号中断

16 ret

17 BootMessage:.ascii "Hello, OS world!"

18 .org 510 #填充到~510~字节处

19 .word 0xaa55 #结束标志

Figure 2.2: boot.S(chapter2/1/boot.S)

2.1.2 用连接脚本控制地址空间

但有一个问题,我们可以使用 nasm boot.asm -o boot.bin命令将 boot.asm直接编译成二进制文
件， GAS不能。不过 GAS的不能恰好给开发者一个机会去分步地实现从汇编源代码到二进制文件这
个过程，使编译更为灵活。下面请看 GAS是如何通过连接脚本控制程序地址空间的：

1 SECTIONS {

2 . = 0x7c00;

3 .text :

4 {

5 _ftext = .; //程序将被加载到内存的7c00处

6 } = 0

7 }

Figure 2.3: boot.S的连接脚本 (chapter2/1/solrex x86.ld)

连接脚本： GNU连接器 ld的每一个连接过程都由连接脚本控制。连接脚本主要用于，怎
样把输入文件内的 section放入输出文件内，并且控制输出文件内各部分在程序地址空间内的
布局。连接器有个默认的内置连接脚本，可以用命令 ld –verbose查看。选项 -T选项可以指
定自己的连接脚本，它将代替默认的连接脚本。

这个连接脚本的功能就是，在连接的时候，将程序入口设置为内存 0x7c00的位置（ BOIS将 PC
控制权转交给这个位置运行的程序），相当于 boot.asm中的 org 07c00h一句。有人可能觉得麻烦，还
需要用一个脚本控制加载地址，但是《自己动手写操作系统》就给了一个很好的反例：《自》第 1.5节
代码 1-2 ,作者切换调试和运行模式时候需要对代码进行注释。

1 ;%define _BOOT_DEBUG_ ; 做 Boot Sector 时一定将此行注释掉!将此行打开后用

2 ; nasm Boot.asm -o Boot.com 做成一个.COM文件易于调试

3

4 %ifdef _BOOT_DEBUG_

5 org 0100h ; 调试状态, 做成 .COM 文件, 可调试

6 %else

7 org 07c00h ; Boot 状态, Bios 将把 Boot Sector 加载到 0:7C00 处并开始执行

8 %endif

9

10 mov ax, cs

11 mov ds, ax

12 mov es, ax

13 call DispStr ; 调用显示字符串例程

14 jmp $ ; 无限循环

15 DispStr:

16 mov ax, BootMessage

17 mov bp, ax ; ES:BP = 串地址

18 mov cx, 16 ; CX = 串长度

19 mov ax, 01301h ; AH = 13, AL = 01h

20 mov bx, 000ch ; 页号为0(BH = 0) 黑底红字(BL = 0Ch,高亮)

21 mov dl, 0

22 int 10h ; 10h 号中断

23 ret

24 BootMessage: db "Hello, OS world!"

25 times 510-($-$$) db 0 ; 填充剩下的空间，使生成的二进制代码恰好为512字节

26 dw 0xaa55 ; 结束标志

Figure 2.4: 《自》代码 1-2 (chapter2/1/boot.asm)

而如果换成使用脚本控制程序地址空间，只需要编译时候调用不同脚本进行连接，就能解决这个问

题。这在嵌入式编程中是很常见的处理方式，即使用不同的连接脚本一次 make从一个源程序文件生成
分别运行在开发板上和软件模拟器上的两个二进制文件。

2.1.3 用Makefile编译连接

下面的这个Makefile文件，就是我们用来自动编译 boot.S汇编源代码的脚本文件：

1 CC=gcc

2 LD=ld

3 LDFILE=solrex_x86.ld #使用上面提供的连接脚本 solrex_x86.ld

4 OBJCOPY=objcopy

5

6 all: boot.img

7

8 # Step 1: gcc 调用 as 将 boot.S 编译成目标文件 boot.o

9 boot.o: boot.S

10 $(CC) -c boot.S

11

12 # Step 2: ld 调用连接脚本 solrex_x86.ld 将 boot.o 连接成可执行文件 boot.elf

13 boot.elf: boot.o

14 $(LD) boot.o -o boot.elf -e c -T$(LDFILE)

15

16 # Step 3: objcopy 移除 boot.elf 中没有用的 section(.pdr,.comment,.note),

17 # strip 掉所有符号信息，输出为二进制文件 boot.bin 。

18 boot.bin : boot.elf

19 @$(OBJCOPY) -R .pdr -R .comment -R.note -S -O binary boot.elf boot.bin

20

21 # Step 4: 生成可启动软盘镜像。

22 boot.img: boot.bin

23 @dd if=boot.bin of=boot.img bs=512 count=1 #用 boot.bin 生成镜像文件第一个扇区

24 # 在 bin 生成的镜像文件后补上空白，最后成为合适大小的软盘镜像

25 @dd if=/dev/zero of=boot.img skip=1 seek=1 bs=512 count=2879

26

27 clean:

28 @rm -rf boot.o boot.elf boot.bin boot.img

Figure 2.5: boot.S的Makefile(chapter2/1/Makefile)

我们将上面内容保存成Makefile，与图 2.2所示 boog.S和图 2.3所示 solrex x86.ld放在同一个目录
下，然后在此目录下使用下面命令编译：

$ make

gcc -c boot.S

ld boot.o -o boot.elf -e c -Tsolrex_x86.ld

ld: warning: cannot find entry symbol c; defaulting to 00007c00

1+0 records in

1+0 records out

512 bytes (512 B) copied, 3.1289e-05 seconds, 16.4 MB/s

2879+0 records in

2879+0 records out

1474048 bytes (1.5 MB) copied, 0.0141508 seconds, 104 MB/s

$ ls

boot.asm boot.elf boot.o Makefile solrex_x86.ld

boot.bin boot.img boot.S solrex.img

可以看到，我们只需执行一条命令 make就可以编译、连接和直接生成可启动的软盘镜像文件，其

间对源文件的每一步处理也都一清二楚。不用任何商业软件，也不用自己写任何转换工具，比如《自己

动手写操作系统》文中提到的 HD-COPY和 Floopy Writer都没有使用到。

在这里需要特别注意的是图 2.5中的 Step 4，其实对这一步的解释应该结合图 1.4来查看。我们用
boot.S编译生成的 boot.bin其实只是图 1.4中所指的软盘的启动扇区，例如 boot.S最后一行：

.word 0xaa55 #结束标志

生成就是启动扇区最后的 0xaa55那两个字节，而 boot.bin的大小是 512字节，正好是启动扇区的大
小。那么 Step 4的功能就是把 boot.bin放入到一个空白软盘的启动扇区，这样呢当虚拟机启动时能识
别出这是一张可启动软盘，并且执行我们在启动扇区中写入的打印代码。

为了验证软盘镜像文件的正确性也可以先用

$ hexdump -x -n 512 boot.img

将 boot.img 前 512 个字节打印出来，可以看到 boot.img dump 的内容和《自》一书附送光盘中的
TINIX.IMG dump的内容完全相同。这里我们也显然用不到 EditPlus或者 UltraEdit，即使需要修改
二进制码，也可以使用 hexedit, ghex2, khexedit等工具对二进制文件进行修改。

下图为使用命令行工具 hexedit打开 boot.img的窗口截图，从图中我们可以看到，左列是该行开头
与文件头对应的偏移地址，中间一列是文件的二进制内容，最右列是文件内容的 ASCII显示内容，可
以看到，此界面与 UltraEdit的十六进制编辑界面没有本质不同。

Figure 2.6: 使用 hexedit打开 boot.img

Figure 2.7: 使用 kde图形界面工具 khexedit打开 boot.img

2.1.4 用虚拟机加载执行 boot.img

当我们生成 boot.img之后，仿照第 1.4.2节中加载软盘镜像的方法，用虚拟机加载 boot.img:

Figure 2.8: 选择启动软盘镜像 boot.img

Figure 2.9: 虚拟机启动后打印出红色的“Hello OS world!”

我们看到虚拟机如我们所料的打印出了红色的“Hello OS world!”字样，这说明我们以上的程序和
编译过程是正确的。

2.2 FAT文件系统

我们在上一节中介绍的内容，仅仅是写一个启动扇区并将其放入软盘镜像的合适位置。由于启动扇

区 512字节的大小限制，我们仅仅能写入像打印一个字符串这样的非常简单的程序，那么如何突破 512
字节的限制呢？很显然的答案是我们要利用其它的扇区，将程序保存在其它扇区，运行前将其加载到内

存后再跳转过去执行。那么又一个问题产生了：程序在软盘上应该怎样存储呢？

可能最直接最容易理解的存储方式就是顺序存储，即将一个大程序从启动扇区开始按顺序存储在相

邻的扇区，可能这样需要的工作量最小，在启动时操作系统仅仅需要序列地将可执行代码拷贝到内存中

来继续运行。可是经过简单的思考我们就可以发现这样做有几个缺陷：1. 软盘中仅能存储操作系统程
序，无法存储其它内容；2. 我们必须使用二进制拷贝方式来制作软盘镜像，修改系统麻烦。

那么怎么避免这两个缺点呢？引入文件系统可以让我们在一张软盘上存储不同的文件，并提供文件

管理功能，可以让我们避免上述的两个缺点。在使用某种文件系统对软盘格式化之后，我们可以像普通

软盘一样使用它来存储多个文件和目录，为了使用软盘上的文件，我们给启动扇区的代码加上寻找文件

和加载执行文件功能，让启动扇区将系统控制权转移给软盘上的某个文件，这样突破启动扇区 512字节
大小的限制。

2.2.1 FAT12文件系统

FAT(File Allocation Table)文件系统规格在 20世纪 70年代末和 80年代初形成，是微软的 MS-
DOS操作系统使用的文件系统格式。它的初衷是为小于 500K容量的软盘制定的简单文件系统，但在

将近三十年的发展过程中，它已经被一次次修改加强以支持更大的存储媒体。在目前主要有三种 FAT
文件系统类型：FAT12, FAT16和 FAT32。这几种类型最基本的区别就像它们的名字字面区别一样，
主要在于大小，即盘上 FAT表的记录项所占的比特数。FAT12的记录项占 12比特，FAT16占 16比
特，FAT32占 32比特。

由于 FAT12最为简单和易实施，这里我们仅简单介绍 FAT12文件系统，想要了解更多 FAT文件
系统知识的话，可以到 http://www.microsoft.com/whdc/system/platform/firmware/fatgen.mspx

下载微软发布的 FAT文件系统官方文档。

FAT12文件系统和其它文件系统一样，都将磁盘划分为层次进行管理。从逻辑上划分，一般将磁
盘划分为分区，目录和文件；从抽象物理结构来讲，将磁盘划分为分区，簇和扇区。那么，如何将逻辑

上的目录和文件映射到物理上实际的簇和扇区，就是文件系统要解决的问题。

如果让虚拟机直接读取我们上一节生成的可启动软盘镜像，或者将 boot.img 软盘用 mount -o

loop boot.img mountdir/挂载到某个目录上，系统肯定会报出“软盘未格式化”或者“文件格式不

可识别”的错误。这是因为任何系统可读取的软盘都是被格式化过的，而我们的 boot.img是一个非常
原始的软盘镜像。那么如何才能使软盘被识别为 FAT12格式的软盘并且可以像普通软盘一样存取呢？

系统在读取一张软盘的时候，会读取软盘上存储的一些关于文件系统的信息，软盘格式化的过程也

就是系统把文件系统信息写入到软盘上的过程。但是我们不能让系统来格式化我们的 boot.img，如果
那样的话，我们写入的启动程序也会被擦除。所以呢，我们需要自己对软盘进行格式化。-可能有人看
到这里就会很沮丧，天那，那该有多麻烦啊！不过我相信在读完以下内容以后你会欢呼雀跃，啊哈，原

来文件系统挺简单的嘛！

2.2.2 启动扇区与 BPB

FAT文件系统的主要信息，都被提供在前几个扇区内，其中第 0号扇区尤其重要。在这个扇区内
隐藏着一个叫做 BPB(BIOS Parameter Block)的数据结构，一旦我们把这个数据结构写对了，格式化
过程也基本完成了,。下面这个表中所示内容，主要就是启动扇区的 BPB数据结构。

表 2.1: 启动扇区的 BPB数据结构和其它内容

名称 偏移 大小 描述 Solrex.img

bytes bytes 文件中的值

BS jmpBoot 0 3 跳转指令，用于过以下的扇区信息 jmp LABEL START

nop

BS OEMName 3 8 厂商名 "WB. YANG"

BPB BytesPerSec 11 2 扇区大小（字节），应为512 512

BPB secPerClus 13 1 簇的扇区数，应为2 的幂，FAT12 为1 1

BPB RsvdSecCnt 14 2 保留扇区，FAT12/16 应为1 1

BPB NumFATs 16 1 FAT 结构数目，一般为2 2

BPB RootEntCnt 17 2 根目录项目数，FAT12 为224 224

BPB TotSec16 19 2 扇区总数，1.44M 软盘为2880 2880

BPB Media 21 1 设备类型，1.44M 软盘为F0h 0xf0

BPB FATSz16 22 2 FAT 占用扇区数，9 9

http://www.microsoft.com/whdc/system/platform/firmware/fatgen.mspx

BPB SecPerTrk 24 2 磁道扇区数，18 18

BPB NumHeads 26 2 磁头数，2 2

BPB HiddSec 28 4 隐藏扇区，默认为0 0

BPB TotSec32 32 4 如果 BPB TotSec16为 0，它记录总扇区数 0

下面的扇区头信息 FAT12/FAT16与 FAT32 不同

BS DrvNum 36 1 中断 0x13的驱动器参数，0为软盘 0

BS Reserved1 37 1 Windows NT 使用，0 0

BS BootSig 38 1 扩展引导标记 (29h)，指明此后 3个域可用 0x29

BS VolID 39 4 卷标序列号，00000000h 0

BS VolLab 43 11 卷标，11 字节，必须用空格20h 补齐 "Solrex 0.01"

BS FilSysType 54 8 文件系统标志，"FAT12 " "FAT12 "

以下为非扇区头信息部分

启动代码及其它 62 448 启动代码、数据及填充字符 mov %cs,%ax...

启动扇区标识符 510 2 可启动扇区标志，0xAA55 0xaa55

哇，天那，这个 BPB看起来很多东西的嘛，怎么写啊？其实写入这些信息很简单，因为它们都是
固定不变的内容，用下面的代码就可以实现。

21 /* Floppy header of FAT12 */

22 jmp LABEL_START /* Start to boot. */

23 nop /* nop required */

24 BS_OEMName: .ascii "WB. YANG" /* OEM String, 8 bytes required */

25 BPB_BytsPerSec: .2byte 512 /* Bytes per sector */

26 BPB_SecPerCluse: .byte 1 /* Sector per cluse */

27 BPB_ResvdSecCnt: .2byte 1 /* Reserved sector count */

28 BPB_NumFATs: .byte 2 /* Number of FATs */

29 BPB_RootEntCnt: .2byte 224 /* Root entries count */

30 BPB_TotSec16: .2byte 2880 /* Total sector number */

31 BPB_Media: .byte 0xf0 /* Media descriptor */

32 BPB_FATSz16: .2byte 9 /* FAT size(sectors) */

33 BPB_SecPerTrk: .2byte 18 /* Sector per track */

34 BPB_NumHeads: .2byte 2 /* Number of magnetic heads */

35 BPB_HiddSec: .4byte 0 /* Number of hidden sectors */

36 BPB_TotSec32: .4byte 0 /* If TotSec16 equal 0, this works */

37 BS_DrvNum: .byte 0 /* Driver number of interrupt 13 */

38 BS_Reserved1: .byte 0 /* Reserved */

39 BS_BootSig: .byte 0x29 /* Boot signal */

40 BS_VolID: .4byte 0 /* Volume ID */

41 BS_VolLab: .ascii "Solrex 0.01" /* Volume label, 11 bytes required */

42 BS_FileSysType: .ascii "FAT12 " /* File system type, 8 bytes required */

43

44 /* Initial registers. */

45 LABEL_START:

Figure 2.10: 生成启动扇区头的汇编代码(节自chapter2/2/boot.S)

在上面的汇编代码中，我们只是顺序地用字符填充了启动扇区头的数据结构，填充的内容与表 2.1
中最后一列的内容相对应。把图 2.10中所示代码添加到图 2.2的第二行和第三行之间，然后再 make，
就能得到一张已经被格式化，可启动也可存储文件的软盘，就是既可以使用 mount -o loop boot.img

mountdir/命令在普通 Linux系统里挂载，也可用作虚拟机启动的软盘镜像文件。

2.2.3 FAT12数据结构

在上一个小节里，我们制作出了可以当作普通软盘使用的启动软盘，这样我们就可以在这张软盘上

存储多个文件了。可还有一步要求我们没有达到，怎样寻找存储的某个引导文件并将其加载到内存中运

行呢？这就涉及到 FAT12文件系统中文件的存储方式了，需要我们了解一些 FAT数据结构和目录结构
的知识。

FAT文件系统对存储空间分配的最小单位是“簇”，因此文件在占用存储空间时，基本单位是簇
而不是字节。即使文件仅仅有 1字节大小，系统也必须分给它一个最小存储单元――簇。由表 2.1中的
BPB secPerClus和 BPB BytsPerSec相乘可以得到每簇所包含的字节数，可见我们设置的是每簇包含
1*512=512个字节，恰好是每簇包含一个扇区。

存储空间分配的最小单位确定了，那么 FAT是如何分配和管理这些存储空间的呢？ FAT的存储空
间管理是通过管理 FAT表来实现的， FAT表一般位于启动扇区之后，根目录之前的若干个扇区，而且
一般是两个表。从根目录区的下一个簇开始，每个簇按照它在磁盘上的位置映射到 FAT表里。FAT文
件系统的存储结构粗略上来讲如图 2.11所示。

Figure 2.11: FAT文件系统存储结构图

FAT表的表项有点儿像数据结构中的单向链表节点的 next指针，先回忆一下，单向链表节点的数
据结构（C语言）是：

struct node {

char * data;

struct node *next;

};

在链表中， next指针指向的是下一个相邻节点。那么 FAT表与链表有什么区别呢？首先， FAT表将
next指针集中管理，放在一起被称为 FAT表；其次， FAT表项指向的是固定大小的文件“簇”(data

段)，而且每个文件簇都有自己对应的 FAT表项。由于每个文件簇都有自己的 FAT表项，这个表项可
能指向另一个文件簇，所以 FAT表项所占字节的多少就决定了 FAT表最大能管理多少内存， FAT12
的 FAT表项有 12个比特，大约能管理 212−个文件簇。
一个文件往往要占据多个簇，只要我们知道这个文件的第一个簇，就可以到 FAT表里查询该簇对

应的 FAT表项，该表项的内容一般就是此文件下一个簇号。如果该表项的值大于 0xff8，则表示该簇

是文件最后一个簇，相当于单向链表节点的 next指针为 NULL；如果该表项的值是 0xff7则表示它是

一个坏簇。这就是文件的链式存储。

2.2.4 FAT12根目录结构

怎样读取一个文件我们知道了，但是如何找到某个文件，即如何得到该文件对应的第一个簇呢？这

就到目录结构派上用场的时候了，为了简单起见，我们这里只介绍根目录结构。

如图 2.11 所示，对于 FAT12/16，根目录存储在磁盘中固定的地方，紧跟在最后一个 FAT 表之
后。根目录的扇区数也是固定的，可以根据 BPB RootEntCnt计算得出：

RootDirSectors = ((BPB_RootEntCnt * 32) + (BPB_BytesPerSec - 1)) / BPB_BytsPerSec

根目录的扇区号是相对于该 FAT卷启动扇区的偏移量：

FirstRootDirSecNum = BPB_RsvdSecCnt + (BPB_NumFATs * BPB_FATSz16)

FAT根目录其实就是一个由 32-bytes的线性表构成的“文件”，其每一个条目代表着一个文件，
这个 32-bytes目录项的格式如图 2.2所示。

表 2.2: 根目录的条目格式

名称 偏移(bytes) 长度(bytes) 描述 举例(loader.bin)

DIR Name 0 0xb 文件名 8字节，扩展名 3字节 "LOADER□□BIN"

DIR Attr 0xb 1 文件属性 0x20

保留位 0xc 10 保留位 0

DIR WrtTime 0x16 2 最后一次写入时间 0x7a5a

DIR WrtDate 0x18 2 最后一次写入日期 0x3188

DIR FstClus 0x1a 2 此目录项的开始簇编号 0x0002

DIR FileSize 0x1c 4 文件大小 0x000000f

知道了这些，我们就得到了足够的信息去在磁盘上寻找某个文件，在磁盘根目录搜索并读取某个文

件的步骤大致如下：

1. 确定根目录区的开始扇区和结束扇区；

2. 遍历根目录区，寻找与被搜索名相对应根目录项；

3. 找到该目录项对应的开始簇编号；

4. 以文件的开始簇为根据寻找整个文件的链接簇，并依次读取每个簇的内容。

2.3 让启动扇区加载引导文件

有了 FAT12文件系统的相关知识之后，我们就可以跨越 512字节的限制，从文件系统中加载文件
并执行了。

2.3.1 一个最简单的 loader

为做测试用，我们写一个最小的程序，让它显示一个字符，然后进入死循环，这样如果 loader加
载成功并成功执行的话，就能看到这个字符。

新建一个文件 loader.S，内容如图 2.12所示。

11 .code16

12 .text

13 mov $0xb800,%ax

14 movw %ax,%gs

15 mov $0xf,%ah

16 mov $’L’,%al

17 mov %ax,%gs:((80*0+39)*2)

18 jmp .

Figure 2.12: 一个最简单的 loader(chapter2/2/loader.S)

这个程序在连接时需要使用连接文件 solrex x86 dos.ld，如图 2.13所示，这样能更改代码段的偏移
量为 0x0100。这样做的目的仅仅是为了与 DOS系统兼容，可以用此代码生成在 DOS下可调试的二进
制文件。

11 SECTIONS

12 {

13 . = 0x0100;

14 .text :

15 {

16 _ftext = .;

17 } = 0

18 }

Figure 2.13: 一个最简单的 loader(chapter2/2/solrex x86 dos.ld)

2.3.2 读取软盘扇区的 BIOS 13h号中断

我们知道了如何在磁盘上寻找一个文件，但是该如何将磁盘上内容读取到内存中去呢？我们在第

2.1节中写的启动扇区不需要自己写代码来读取，是因为它每次都被加载到内存的固定位置，计算机在
发现可启动标识 0xaa55的时候自动就会做加载工作。但如果我们想自己从软盘上读取文件的时候，就

需要使用到底层 BIOS系统提供的磁盘读取功能了。这里，我们主要用到 BIOS 13h号中断。

表 2.3所示，就是 BIOS 13号中断的参数表。从表中我们可以看到，读取磁盘驱动器所需要的参数
是磁道（柱面）号、磁头号以及当前磁道上的扇区号三个分量。由第 1.2节所介绍的磁盘知识，我们可
以得到计算这三个分量的公式 2.1。

扇区号

18(每磁道扇区数)
=





商 Q =




柱面号 = Q >> 1

磁头号 = Q & 1

余数 R =⇒ 起始扇区号 = R + 1

(2.1)

表 2.3: BIOS 13h号中断的参数表

中断号 AH 功能 调用参数 返回参数

13

0 磁盘复位 DL = 驱动器号 失败：

00, 01 为软盘，80h, 81h，· · · 为硬盘 AH = 错误码

1 读磁盘驱动器状态 AH=状态字节

2 读磁盘扇区 AL = 扇区数 读成功：

(CL)6,7(CH)0∼7 =磁盘号 AH = 0

(CL)0∼5 =扇区号 AL = 读取的扇区数

DH/DL = 磁头号/驱动器号 读失败：

ES:BX = 数据缓冲区地址 AH = 错误码

3 写磁盘扇区 同上 写成功：

AH = 0

AL = 写入的扇区数

写失败：

AH = 错误码

4 检验磁盘扇区 AL = 扇区数 成功：

(CL)6,7(CH)0∼7 =磁盘号 AH = 0

(CL)0∼5 =扇区号 AL = 检验的扇区数

DH/DL = 磁头号/驱动器号 失败：AH = 错误码

5 格式化盘磁道 AL = 扇区数 成功：

(CL)6,7(CH)0∼7 =磁盘号 AH = 0

(CL)0∼5 =扇区号 失败：

DH/DL = 磁头号/驱动器号 AH = 错误码

ES:BX = 格式化参数表指针

知道了这些，我们就可以写一个读取软盘扇区的子函数了：

209 /* ==================================================================

210 Routine: ReadSector

211 Action: Read %cl Sectors from %ax sector(floppy) to %es:%bx(memory)

212 Assume sector number is ’x’, then:

213 x/(BPB_SecPerTrk) = y,

214 x%(BPB_SecPerTrk) = z.

215 The remainder ’z’ PLUS 1 is the start sector number;

216 The quotient ’y’ devide by BPB_NumHeads(RIGHT SHIFT 1 bit)is cylinder

217 number;

218 AND ’y’ by 1 can got magnetic header.

219 */

220 ReadSector:

221 push %ebp

222 mov %esp,%ebp

223 sub $2,%esp /* Reserve space for saving %cl */

224 mov %cl,-2(%ebp)

225 push %bx /* Save bx */

226 mov (BPB_SecPerTrk), %bl /* %bl: the devider */

227 div %bl /* ’y’ in %al, ’z’ in %ah */

228 inc %ah /* z++, got start sector */

229 mov %ah,%cl /* %cl <- start sector number */

230 mov %al,%dh /* %dh <- ’y’ */

231 shr $1,%al /* ’y’/BPB_NumHeads */

232 mov %al,%ch /* %ch <- Cylinder number(y>>1) */

233 and $1,%dh /* %dh <- Magnetic header(y&1) */

234 pop %bx /* Restore %bx */

235 /* Now, we got cylinder number in %ch, start sector number in %cl, magnetic

236 header in %dh. */

237 mov (BS_DrvNum), %dl

238 GoOnReading:

239 mov $2,%ah

240 mov -2(%ebp),%al /* Read %al sectors */

241 int $0x13

242 jc GoOnReading /* If CF set 1, mean read error, reread. */

243 add $2,%esp

244 pop %ebp

245 ret

246

Figure 2.14: 读取软盘扇区的函数(节自chapter2/2/boot.S)

2.3.3 搜索 loader.bin

读取扇区的子函数写好了，下面我们编写在软盘中搜索 loader.bin的代码：

62

63 /* Reset floppy */

64 xor %ah,%ah

65 xor %dl,%dl /* %dl=0: floppy driver 0 */

66 int $0x13 /* BIOS int 13h, ah=0: Reset driver 0 */

67

68 /* Find LOADER.BIN in root directory of driver 0 */

69 movw $SecNoOfRootDir, (wSectorNo)

70

71 /* Read root dir sector to memory */

72 LABEL_SEARCH_IN_ROOT_DIR_BEGIN:

73 cmpw $0,(wRootDirSizeForLoop) /* If searching in root dir */

74 jz LABEL_NO_LOADERBIN /* can find LOADER.BIN ? */

75 decw (wRootDirSizeForLoop)

76 mov $BaseOfLoader,%ax

77 mov %ax,%es /* %es <- BaseOfLoader*/

78 mov $OffsetOfLoader,%bx /* %bx <- OffsetOfLoader */

79 mov (wSectorNo),%ax /* %ax <- sector number in root */

80 mov $1,%cl

81 call ReadSector

82 mov $LoaderFileName,%si /* %ds:%si -> LOADER BIN */

83 mov $OffsetOfLoader,%di /* BaseOfLoader<<4+100*/

84 cld

85 mov $0x10,%dx

86

87 /* Search for "LOADER BIN", FAT12 save file name in 12 bytes, 8 bytes for

88 file name, 3 bytes for suffix, last 1 bytes for ’\20’. If file name is

89 less than 8 bytes, filled with ’\20’. So "LOADER.BIN" is saved as:

90 "LOADER BIN"(4f4c 4441 5245 2020 4942 204e).

91 */

92 LABEL_SEARCH_FOR_LOADERBIN:

93 cmp $0,%dx /* Read control */

94 jz LABEL_GOTO_NEXT_SECTOR_IN_ROOT_DIR

95 dec %dx

96 mov $11,%cx

97

98 LABEL_CMP_FILENAME:

99 cmp $0,%cx

100 jz LABEL_FILENAME_FOUND /* If 11 chars are all identical? */

101 dec %cx

102 lodsb /* %ds:(%si) -> %al*/

103 cmp %es:(%di),%al

104 jz LABEL_GO_ON

105 jmp LABEL_DIFFERENT /* Different */

106

107 LABEL_GO_ON:

108 inc %di

109 jmp LABEL_CMP_FILENAME /* Go on loop */

110

111 LABEL_DIFFERENT:

112 and $0xffe0,%di /* Go to head of this entry */

113 add $0x20,%di

114 mov $LoaderFileName,%si /* Next entry */

115 jmp LABEL_SEARCH_FOR_LOADERBIN

116

117 LABEL_GOTO_NEXT_SECTOR_IN_ROOT_DIR:

118 addw $1,(wSectorNo)

119 jmp LABEL_SEARCH_IN_ROOT_DIR_BEGIN

120

121 /* Not found LOADER.BIN in root dir. */

122 LABEL_NO_LOADERBIN:

123 mov $2,%dh

124 call DispStr /* Display string(index 2) */

125 jmp . /* Infinite loop */

126

127 /* Found. */

128 LABEL_FILENAME_FOUND:

129 mov $RootDirSectors,%ax

130 and $0xffe0,%di /* Start of current entry, 32 bytes per entry */

131 add $0x1a,%di /* First sector of this file */

132 mov %es:(%di),%cx

133 push %cx /* Save index of this sector in FAT */

134 add %ax,%cx

135 add $DeltaSecNo,%cx /* LOADER.BIN’s start sector saved in %cl */

136 mov $BaseOfLoader,%ax

137 mov %ax,%es /* %es <- BaseOfLoader */

138 mov $OffsetOfLoader,%bx /* %bx <- OffsetOfLoader */

139 mov %cx,%ax /* %ax <- Sector number */

140

Figure 2.15: 搜索 loader.bin的代码片段(节自chapter2/2/boot.S)

这段代码的功能就是我们前面提到过的，遍历根目录的所有扇区，将每个扇区加载入内存，然后从

中寻找文件名为 loader.bin的条目，直到找到为止。找到之后，计算出 loader.bin的起始扇区号。其中
用到的变量和字符串的定义见图 2.16中代码片段的定义。

12 .set BaseOfStack, 0x7c00 /* Stack base address, inner */

13 .set BaseOfLoader, 0x9000 /* Section loading address of LOADER.BIN */

14 .set OffsetOfLoader, 0x0100 /* Loading offset of LOADER.BIN */

15 .set RootDirSectors, 14 /* Root directory sector count */

16 .set SecNoOfRootDir, 19 /* 1st sector of root directory */

17 .set SecNoOfFAT1, 1 /* 1st sector of FAT1 */

18 .set DeltaSecNo, 17 /* BPB_(RsvdSecCnt+NumFATs*FATSz) -2 */

174

175 /* ==================================================================

176 Variable table

177 */

178 wRootDirSizeForLoop: .2byte RootDirSectors

179 wSectorNo: .2byte 0 /* Sector number to read */

180 bOdd: .byte 0 /* odd or even? */

181

182 /* ==================================================================

183 String table

184 */

185 LoaderFileName: .asciz "LOADER BIN" /* File name */

186 .set MessageLength,9

187 BootMessage: .ascii "Booting**" /* index 0 */

188 Message1: .ascii "Loaded in" /* index 1 */

189 Message2: .ascii "No LOADER" /* index 2 */

190

Figure 2.16: 搜索 loader.bin使用的变量定义(节自chapter2/2/boot.S)

由于在代码中有一些打印工作，我们写了一个函数专门做这项工作。为了节省代码长度，被打印字

符串的长度都设置为 9字节，不够则用空格补齐，这样就相当于一个备用的二维数组，通过数字定位要

打印的字符串，很方便。打印字符串的函数 DispStr见图 2.17，调用它的时候需要从寄存器 dh传入参
数字符串序号。

190

191 /* ==================================================================

192 Routine: DispStr

193 Action: Display a string, string index stored in %dh

194 */

195 DispStr:

196 mov $MessageLength, %ax

197 mul %dh

198 add $BootMessage,%ax

199 mov %ax,%bp /* String address */

200 mov %ds,%ax

201 mov %ax,%es

202 mov $MessageLength,%cx /* String length */

203 mov $0x1301,%ax /* ah = 0x13, al = 0x01(W) */

204 mov $0x07,%bx /* PageNum 0(bh = 0), bw(bl= 0x07)*/

205 mov $0,%dl /* Start row and column */

206 int $0x10 /* BIOS INT 10h, display string */

207 ret

208

Figure 2.17: 打印字符串函数 DispStr (节自chapter2/2/boot.S)

2.3.4 加载 loader入内存

在寻找到 loader.bin之后，就需要把它装入内存。现在我们已经有了 loader.bin的起始扇区号，利
用这个扇区号可以做两件事：一，把起始扇区装入内存；二，通过它找到 FAT中的条目，从而找到
loader.bin文件所占用的其它扇区。

这里，我们把 loader.bin装入内存中的 BaseOfLoader:OffsetOfLoader处，但是在图 2.15中我们将
根目录区也是装载到这个位置。因为在找到 loader.bin之后，该内存区域对我们已经没有用处了，所以
它尽可以被覆盖。

我们已经知道了如何装入一个扇区，但是从 FAT表中寻找其它的扇区还是意见麻烦的事情，所以
我们写了一个函数 GetFATEntry来专门做这件事情，函数的输入是扇区号，输出是其对应的 FAT项的
值，见图 2.18。

246

247 /* ==================================================================

248 Routine: GetFATEntry

249 Action: Find %ax sector’s index in FAT, save result in %ax

250 */

251 GetFATEntry:

252 push %es

253 push %bx

254 push %ax

255 mov $BaseOfLoader,%ax

256 sub $0x0100,%ax

257 mov %ax,%es /* Left 4K bytes for FAT */

258 pop %ax

259 movb $0,(bOdd)

260 mov $3,%bx

261 mul %bx /* %dx:%ax = %ax*3 */

262 mov $2,%bx

263 div %bx /* %dx:%ax/2 */

264 cmp $0,%bx /* remainder %dx = 0 ? */

265 jz LABEL_EVEN

266 movb $1,(bOdd)

267

268 LABEL_EVEN:

269 xor %dx,%dx /* Now %ax is the offset of FATEntry in FAT */

270 mov (BPB_BytsPerSec),%bx

271 div %bx /* %dx:%ax/BPB_BytsPerSec */

272 push %dx

273 mov $0,%bx

274 add $SecNoOfFAT1,%ax /* %ax <- FATEntry’s sector */

275 mov $2,%cl /* Read 2 sectors in 1 time, because FATEntry */

276 call ReadSector /* may be in 2 sectors. */

277 pop %dx

278 add %dx,%bx

279 mov %es:(%bx),%ax

280 cmpb $1,(bOdd)

281 jnz LABEL_EVEN_2

282 shr $4,%ax

283

284 LABEL_EVEN_2:

285 and $0x0fff,%ax

286

287 LABEL_GET_FAT_ENTRY_OK:

288 pop %bx

289 pop %es

290 ret

291

Figure 2.18: 寻找 FAT项的函数 GetFATEntry (节自chapter2/2/boot.S)

这样，我们就可以将 loader.bin整个文件加载到内存中去了，见图 2.19。

140

141 /* Load LOADER.BIN’s sector’s to memory. */

142 LABEL_GOON_LOADING_FILE:

143 push %ax

144 push %bx

145 mov $0x0e,%ah

146 mov $’.’,%al /* Char to print */

147 mov $0x0f,%bl /* Front color: white */

148 int $0x10 /* BIOS int 10h, ah=0xe: Print char */

149 pop %bx

150 pop %ax

151

152 mov $1,%cl

153 call ReadSector

154 pop %ax /* Got index of this sector in FAT */

155 call GetFATEntry

156 cmp $0x0fff,%ax

157 jz LABEL_FILE_LOADED

158 push %ax /* Save index of this sector in FAT */

159 mov $RootDirSectors,%dx

160 add %dx,%ax

161 add $DeltaSecNo,%ax

162 add (BPB_BytsPerSec),%bx

163 jmp LABEL_GOON_LOADING_FILE

164

165 LABEL_FILE_LOADED:

166 mov $1,%dh

167 call DispStr /* Display string(index 1) */

168

Figure 2.19: 加载 loader.bin的代码(节自chapter2/2/boot.S)

在图 2.19中我们看到一个宏 DeltaSectorNo，这个宏就是为了将 FAT中的簇号转换为扇区号。由
于根目录区的开始扇区号是 19，而 FAT表的前两个项 0,1分别是磁盘识别字和被保留，其表项其实是
从第 2项开始的，第 2项对应着根目录区后的第一个扇区，所以扇区号和簇号的对应关系就是：

扇区号 = 簇号 +根目录区占用扇区数 +根目录区开始扇区号 - 2

= 簇号 +根目录区占用扇区数 + 17

这就是 DeltaSectorNo的值 17的由来。

2.3.5 向 loader转交控制权

我们已经将 loader成功地加载入了内存，然后就需要进行一个跳转，来执行 loader。

168

169 /*******************************************************************

170 Jump to LOADER.BIN’s start address in memory.

171 */

172 jmp $BaseOfLoader,$OffsetOfLoader

173 /*******************************************************************/

174

Figure 2.20: 跳转到 loader执行(节自chapter2/2/boot.S)

2.3.6 生成镜像并测试

我们写好了汇编源代码，那么就需要将源代码编译成可执行文件，并生成光盘镜像了。

11 CC=gcc

12 LD=ld

13 OBJCOPY=objcopy

14

15 CFLAGS=-c

16 TRIM_FLAGS=-R .pdr -R .comment -R.note -S -O binary

17

18 LDFILE_BOOT=solrex_x86_boot.ld

19 LDFILE_DOS=solrex_x86_dos.ld

20 LDFLAGS_BOOT=-e c -T$(LDFILE_BOOT)

21 LDFLAGS_DOS=-e c -T$(LDFILE_DOS)

22

23 all: boot.img LOADER.BIN

24

25 boot.bin: boot.S

26 $(CC) $(CFLAGS) boot.S

27 $(LD) boot.o -o boot.elf $(LDFLAGS_BOOT)

28 $(OBJCOPY) $(TRIM_FLAGS) boot.elf $@

29

30 LOADER.BIN: loader.S

31 $(CC) $(CFLAGS) loader.S

32 $(LD) loader.o -o loader.elf $(LDFLAGS_DOS)

33 $(OBJCOPY) $(TRIM_FLAGS) loader.elf $@

34

35 boot.img: boot.bin

36 dd if=boot.bin of=boot.img bs=512 count=1

37 dd if=/dev/zero of=boot.img skip=1 seek=1 bs=512 count=2879

38

Figure 2.21: 用Makefile编译(节自chapter2/2/boot.S)

上面的代码比较简单，我们可以通过一个make命令编译生成 boot.img和 LOADER.BIN：

$ make

gcc -c boot.S

boot.S:293:67: warning: no newline at end of file

ld boot.o -o boot.elf -e c -Tsolrex_x86_boot.ld

ld: warning: cannot find entry symbol c; defaulting to 00007c00

objcopy -R .pdr -R .comment -R.note -S -O binary boot.elf boot.bin

dd if=boot.bin of=boot.img bs=512 count=1

1+0 records in

1+0 records out

512 bytes (512 B) copied, 6.5651e-05 seconds, 7.8 MB/s

dd if=/dev/zero of=boot.img skip=1 seek=1 bs=512 count=2879

2879+0 records in

2879+0 records out

1474048 bytes (1.5 MB) copied, 0.0305594 seconds, 48.2 MB/s

gcc -c loader.S

ld loader.o -o loader.elf -e c -Tsolrex_x86_dos.ld

ld: warning: cannot find entry symbol c; defaulting to 00000100

objcopy -R .pdr -R .comment -R.note -S -O binary loader.elf LOADER.BIN

由于我们的目标就是让启动扇区加载引导文件，所以需要把引导文件放入软盘镜像中。那么如何

将 LOADER.BIN放入 boot.img中呢？我们只需要挂载 boot.img并将 LOADER.BIN拷贝进入被挂载
的目录，为此我们在Makefile中添加新的编译目标 copy：

38

39 # You must have the authority to do mount, or you must use "su root" or

40 # "sudo" command to do "make copy"

41 copy: boot.img LOADER.BIN

42 mkdir -p /tmp/floppy;\

43 mount -o loop boot.img /tmp/floppy/ -o fat=12;\

44 cp LOADER.BIN /tmp/floppy/;\

45 umount /tmp/floppy/;\

46 rm -rf /tmp/floppy/;

47

Figure 2.22: 拷贝 LOADER.BIN入 boot.img(节自chapter2/2/boot.S)

由于挂载软盘镜像在很多 Linux系统上需要 root权限，所以我们没有将 copy目标添加到 all的依
赖关系中。在执行make copy命令之前我们必须先获得 root权限。

$ su

Password:

# make copy

mkdir -p /tmp/floppy;\

mount -o loop boot.img /tmp/floppy/ -o fat=12;\

cp LOADER.BIN /tmp/floppy/;\

umount /tmp/floppy/;\

rm -rf /tmp/floppy/;

如果仅仅生成 boot.img而不将 loader.bin装入它，用这样的软盘启动会显示找不到 LOADER：

Figure 2.23: 没有装入 LOADER.BIN的软盘启动

装入 loader.bin之后再用 boot.img启动，我们看到虚拟机启动并在屏幕中间打印出了一个字符 ”L”
，这说明我们前面的工作都是正确的。

Figure 2.24: 装入了 LOADER.BIN以后再启动

2.4 本章全部源代码

为了避免本书官方网站发生问题，所以我们特别将每章的源代码附在每章最后一节，以方便读者使

用。

1 /* chapter2/1/boot.S

2

3 Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4

5 This file is part of the source code of book "Write Operating System with

6 Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7

8 This file is licensed under the GNU General Public License; either

9 version 3 of the License, or (at your option) any later version. */

10

11 .code16

12 .text

13 mov %cs,%ax

14 mov %ax,%ds

15 mov %ax,%es

16 call DispStr

17 jmp .

18 DispStr:

19 mov $BootMessage, %ax

20 mov %ax,%bp

21 mov $16,%cx

22 mov $0x1301,%ax

23 mov $0x00c,%bx

24 mov $0,%dl

25 int $0x10

26 ret

27 BootMessage:.ascii "Hello, OS world!"

28 .org 510

29 .word 0xaa55

Figure 2.25: chapter2/1/boot.S

1 /* chapter2/2/boot.S

2

3 Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4

5 This file is part of the source code of book "Write Operating System with

6 Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7

8 This file is licensed under the GNU General Public License; either

9 version 3 of the License, or (at your option) any later version. */

10

11 SECTIONS

12 {

13 . = 0x7c00;

14 .text :

15 {

16 _ftext = .;

17 } = 0

18 }

Figure 2.26: chapter2/1/solrex x86.ld

1 # chapter2/1/Makefile

2 #

3 # Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4 #

5 # This file is part of the source code of book "Write Operating System with

6 # Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7 #

8 # This file is licensed under the GNU General Public License; either

9 # version 3 of the License, or (at your option) any later version.

10

11 CC=gcc

12 LD=ld

13 LDFILE=solrex_x86.ld

14 OBJCOPY=objcopy

15

16 all: boot.img

17

18 boot.img: boot.bin

19 @dd if=boot.bin of=boot.img bs=512 count=1

20 @dd if=/dev/zero of=boot.img skip=1 seek=1 bs=512 count=2879

21

22 boot.bin: boot.elf

23 @$(OBJCOPY) -R .pdr -R .comment -R.note -S -O binary boot.elf boot.bin

24

25 boot.elf: boot.o

26 $(LD) boot.o -o boot.elf -e c -Tsolrex_x86.ld

27

28 boot.o: boot.S

29 $(CC) -c boot.S

30

31 clean:

32 @rm -f boot.o boot.elf boot.bin

33

34 distclean:

35 @rm -f *.img

Figure 2.27: chapter2/1/Makefile

1 /* chapter2/2/boot.S

2

3 Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4

5 This file is part of the source code of book "Write Operating System with

6 Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7

8 This file is licensed under the GNU General Public License; either

9 version 3 of the License, or (at your option) any later version. */

10

11 .code16

12 .set BaseOfStack, 0x7c00 /* Stack base address, inner */

13 .set BaseOfLoader, 0x9000 /* Section loading address of LOADER.BIN */

14 .set OffsetOfLoader, 0x0100 /* Loading offset of LOADER.BIN */

15 .set RootDirSectors, 14 /* Root directory sector count */

16 .set SecNoOfRootDir, 19 /* 1st sector of root directory */

17 .set SecNoOfFAT1, 1 /* 1st sector of FAT1 */

18 .set DeltaSecNo, 17 /* BPB_(RsvdSecCnt+NumFATs*FATSz) -2 */

19 /* Start sector of file space =*/

20 .text

21 /* Floppy header of FAT12 */

22 jmp LABEL_START /* Start to boot. */

23 nop /* nop required */

24 BS_OEMName: .ascii "WB. YANG" /* OEM String, 8 bytes required */

25 BPB_BytsPerSec: .2byte 512 /* Bytes per sector */

26 BPB_SecPerCluse: .byte 1 /* Sector per cluse */

27 BPB_ResvdSecCnt: .2byte 1 /* Reserved sector count */

28 BPB_NumFATs: .byte 2 /* Number of FATs */

29 BPB_RootEntCnt: .2byte 224 /* Root entries count */

30 BPB_TotSec16: .2byte 2880 /* Total sector number */

31 BPB_Media: .byte 0xf0 /* Media descriptor */

32 BPB_FATSz16: .2byte 9 /* FAT size(sectors) */

33 BPB_SecPerTrk: .2byte 18 /* Sector per track */

34 BPB_NumHeads: .2byte 2 /* Number of magnetic heads */

35 BPB_HiddSec: .4byte 0 /* Number of hidden sectors */

36 BPB_TotSec32: .4byte 0 /* If TotSec16 equal 0, this works */

37 BS_DrvNum: .byte 0 /* Driver number of interrupt 13 */

38 BS_Reserved1: .byte 0 /* Reserved */

39 BS_BootSig: .byte 0x29 /* Boot signal */

40 BS_VolID: .4byte 0 /* Volume ID */

41 BS_VolLab: .ascii "Solrex 0.01" /* Volume label, 11 bytes required */

42 BS_FileSysType: .ascii "FAT12 " /* File system type, 8 bytes required */

43

44 /* Initial registers. */

45 LABEL_START:

46 mov %cs,%ax

47 mov %ax,%ds

48 mov %ax,%es

49 mov %ax,%ss

50 mov $BaseOfStack, %sp

51

52 /* Clear screen */

53 mov $0x0600,%ax /* %ah=6, %al=0 */

54 mov $0x0700,%bx /* Black white */

55 mov $0,%cx /* Top left: (0,0) */

56 mov $0x184f,%dx /* Bottom right: (80,50) */

57 int $0x10 /* BIOS int 10h, ah=6: Initialize screen */

58

59 /* Display "Booting**" */

60 mov $0,%dh

61 call DispStr /* Display string(index 0)*/

62

63 /* Reset floppy */

64 xor %ah,%ah

65 xor %dl,%dl /* %dl=0: floppy driver 0 */

66 int $0x13 /* BIOS int 13h, ah=0: Reset driver 0 */

67

68 /* Find LOADER.BIN in root directory of driver 0 */

69 movw $SecNoOfRootDir, (wSectorNo)

70

71 /* Read root dir sector to memory */

72 LABEL_SEARCH_IN_ROOT_DIR_BEGIN:

73 cmpw $0,(wRootDirSizeForLoop) /* If searching in root dir */

74 jz LABEL_NO_LOADERBIN /* can find LOADER.BIN ? */

75 decw (wRootDirSizeForLoop)

76 mov $BaseOfLoader,%ax

77 mov %ax,%es /* %es <- BaseOfLoader*/

78 mov $OffsetOfLoader,%bx /* %bx <- OffsetOfLoader */

79 mov (wSectorNo),%ax /* %ax <- sector number in root */

80 mov $1,%cl

81 call ReadSector

82 mov $LoaderFileName,%si /* %ds:%si -> LOADER BIN */

83 mov $OffsetOfLoader,%di /* BaseOfLoader<<4+100*/

84 cld

85 mov $0x10,%dx

86

87 /* Search for "LOADER BIN", FAT12 save file name in 12 bytes, 8 bytes for

88 file name, 3 bytes for suffix, last 1 bytes for ’\20’. If file name is

89 less than 8 bytes, filled with ’\20’. So "LOADER.BIN" is saved as:

90 "LOADER BIN"(4f4c 4441 5245 2020 4942 204e).

91 */

92 LABEL_SEARCH_FOR_LOADERBIN:

93 cmp $0,%dx /* Read control */

94 jz LABEL_GOTO_NEXT_SECTOR_IN_ROOT_DIR

95 dec %dx

96 mov $11,%cx

97

98 LABEL_CMP_FILENAME:

99 cmp $0,%cx

100 jz LABEL_FILENAME_FOUND /* If 11 chars are all identical? */

101 dec %cx

102 lodsb /* %ds:(%si) -> %al*/

103 cmp %es:(%di),%al

104 jz LABEL_GO_ON

105 jmp LABEL_DIFFERENT /* Different */

106

107 LABEL_GO_ON:

108 inc %di

109 jmp LABEL_CMP_FILENAME /* Go on loop */

110

111 LABEL_DIFFERENT:

112 and $0xffe0,%di /* Go to head of this entry */

113 add $0x20,%di

114 mov $LoaderFileName,%si /* Next entry */

115 jmp LABEL_SEARCH_FOR_LOADERBIN

116

117 LABEL_GOTO_NEXT_SECTOR_IN_ROOT_DIR:

118 addw $1,(wSectorNo)

119 jmp LABEL_SEARCH_IN_ROOT_DIR_BEGIN

120

121 /* Not found LOADER.BIN in root dir. */

122 LABEL_NO_LOADERBIN:

123 mov $2,%dh

124 call DispStr /* Display string(index 2) */

125 jmp . /* Infinite loop */

126

127 /* Found. */

128 LABEL_FILENAME_FOUND:

129 mov $RootDirSectors,%ax

130 and $0xffe0,%di /* Start of current entry, 32 bytes per entry */

131 add $0x1a,%di /* First sector of this file */

132 mov %es:(%di),%cx

133 push %cx /* Save index of this sector in FAT */

134 add %ax,%cx

135 add $DeltaSecNo,%cx /* LOADER.BIN’s start sector saved in %cl */

136 mov $BaseOfLoader,%ax

137 mov %ax,%es /* %es <- BaseOfLoader */

138 mov $OffsetOfLoader,%bx /* %bx <- OffsetOfLoader */

139 mov %cx,%ax /* %ax <- Sector number */

140

141 /* Load LOADER.BIN’s sector’s to memory. */

142 LABEL_GOON_LOADING_FILE:

143 push %ax

144 push %bx

145 mov $0x0e,%ah

146 mov $’.’,%al /* Char to print */

147 mov $0x0f,%bl /* Front color: white */

148 int $0x10 /* BIOS int 10h, ah=0xe: Print char */

149 pop %bx

150 pop %ax

151

152 mov $1,%cl

153 call ReadSector

154 pop %ax /* Got index of this sector in FAT */

155 call GetFATEntry

156 cmp $0x0fff,%ax

157 jz LABEL_FILE_LOADED

158 push %ax /* Save index of this sector in FAT */

159 mov $RootDirSectors,%dx

160 add %dx,%ax

161 add $DeltaSecNo,%ax

162 add (BPB_BytsPerSec),%bx

163 jmp LABEL_GOON_LOADING_FILE

164

165 LABEL_FILE_LOADED:

166 mov $1,%dh

167 call DispStr /* Display string(index 1) */

168

169 /*******************************************************************

170 Jump to LOADER.BIN’s start address in memory.

171 */

172 jmp $BaseOfLoader,$OffsetOfLoader

173 /*******************************************************************/

174

175 /* ==================================================================

176 Variable table

177 */

178 wRootDirSizeForLoop: .2byte RootDirSectors

179 wSectorNo: .2byte 0 /* Sector number to read */

180 bOdd: .byte 0 /* odd or even? */

181

182 /* ==================================================================

183 String table

184 */

185 LoaderFileName: .asciz "LOADER BIN" /* File name */

186 .set MessageLength,9

187 BootMessage: .ascii "Booting**" /* index 0 */

188 Message1: .ascii "Loaded in" /* index 1 */

189 Message2: .ascii "No LOADER" /* index 2 */

190

191 /* ==================================================================

192 Routine: DispStr

193 Action: Display a string, string index stored in %dh

194 */

195 DispStr:

196 mov $MessageLength, %ax

197 mul %dh

198 add $BootMessage,%ax

199 mov %ax,%bp /* String address */

200 mov %ds,%ax

201 mov %ax,%es

202 mov $MessageLength,%cx /* String length */

203 mov $0x1301,%ax /* ah = 0x13, al = 0x01(W) */

204 mov $0x07,%bx /* PageNum 0(bh = 0), bw(bl= 0x07)*/

205 mov $0,%dl /* Start row and column */

206 int $0x10 /* BIOS INT 10h, display string */

207 ret

208

209 /* ==================================================================

210 Routine: ReadSector

211 Action: Read %cl Sectors from %ax sector(floppy) to %es:%bx(memory)

212 Assume sector number is ’x’, then:

213 x/(BPB_SecPerTrk) = y,

214 x%(BPB_SecPerTrk) = z.

215 The remainder ’z’ PLUS 1 is the start sector number;

216 The quotient ’y’ devide by BPB_NumHeads(RIGHT SHIFT 1 bit)is cylinder

217 number;

218 AND ’y’ by 1 can got magnetic header.

219 */

220 ReadSector:

221 push %ebp

222 mov %esp,%ebp

223 sub $2,%esp /* Reserve space for saving %cl */

224 mov %cl,-2(%ebp)

225 push %bx /* Save bx */

226 mov (BPB_SecPerTrk), %bl /* %bl: the devider */

227 div %bl /* ’y’ in %al, ’z’ in %ah */

228 inc %ah /* z++, got start sector */

229 mov %ah,%cl /* %cl <- start sector number */

230 mov %al,%dh /* %dh <- ’y’ */

231 shr $1,%al /* ’y’/BPB_NumHeads */

232 mov %al,%ch /* %ch <- Cylinder number(y>>1) */

233 and $1,%dh /* %dh <- Magnetic header(y&1) */

234 pop %bx /* Restore %bx */

235 /* Now, we got cylinder number in %ch, start sector number in %cl, magnetic

236 header in %dh. */

237 mov (BS_DrvNum), %dl

238 GoOnReading:

239 mov $2,%ah

240 mov -2(%ebp),%al /* Read %al sectors */

241 int $0x13

242 jc GoOnReading /* If CF set 1, mean read error, reread. */

243 add $2,%esp

244 pop %ebp

245 ret

246

247 /* ==================================================================

248 Routine: GetFATEntry

249 Action: Find %ax sector’s index in FAT, save result in %ax

250 */

251 GetFATEntry:

252 push %es

253 push %bx

254 push %ax

255 mov $BaseOfLoader,%ax

256 sub $0x0100,%ax

257 mov %ax,%es /* Left 4K bytes for FAT */

258 pop %ax

259 movb $0,(bOdd)

260 mov $3,%bx

261 mul %bx /* %dx:%ax = %ax*3 */

262 mov $2,%bx

263 div %bx /* %dx:%ax/2 */

264 cmp $0,%bx /* remainder %dx = 0 ? */

265 jz LABEL_EVEN

266 movb $1,(bOdd)

267

268 LABEL_EVEN:

269 xor %dx,%dx /* Now %ax is the offset of FATEntry in FAT */

270 mov (BPB_BytsPerSec),%bx

271 div %bx /* %dx:%ax/BPB_BytsPerSec */

272 push %dx

273 mov $0,%bx

274 add $SecNoOfFAT1,%ax /* %ax <- FATEntry’s sector */

275 mov $2,%cl /* Read 2 sectors in 1 time, because FATEntry */

276 call ReadSector /* may be in 2 sectors. */

277 pop %dx

278 add %dx,%bx

279 mov %es:(%bx),%ax

280 cmpb $1,(bOdd)

281 jnz LABEL_EVEN_2

282 shr $4,%ax

283

284 LABEL_EVEN_2:

285 and $0x0fff,%ax

286

287 LABEL_GET_FAT_ENTRY_OK:

288 pop %bx

289 pop %es

290 ret

291

292 .org 510 /* Skip to address 0x510. */

293 .2byte 0xaa55 /* Write boot flag to 1st sector(512 bytes) end */

Figure 2.28: chapter2/2/boot.S

1 /* chapter2/2/loader.S

2

3 Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4

5 This file is part of the source code of book "Write Operating System with

6 Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7

8 This file is licensed under the GNU General Public License; either

9 version 3 of the License, or (at your option) any later version. */

10

11 .code16

12 .text

13 mov $0xb800,%ax

14 movw %ax,%gs

15 mov $0xf,%ah

16 mov $’L’,%al

17 mov %ax,%gs:((80*0+39)*2)

18 jmp .

Figure 2.29: chapter2/2/loader.S

1 /* chapter2/2/solrex_x86_boot.ld

2

3 Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4

5 This file is part of the source code of book "Write Operating System with

6 Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7

8 This file is licensed under the GNU General Public License; either

9 version 3 of the License, or (at your option) any later version. */

10

11 SECTIONS

12 {

13 . = 0x7c00;

14 .text :

15 {

16 _ftext = .;

17 } = 0

18 }

Figure 2.30: chapter2/2/solrex x86 boot.ld

1 /* chapter2/2/solrex_x86_dos.ld

2

3 Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4

5 This file is part of the source code of book "Write Operating System with

6 Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7

8 This file is licensed under the GNU General Public License; either

9 version 3 of the License, or (at your option) any later version. */

10

11 SECTIONS

12 {

13 . = 0x0100;

14 .text :

15 {

16 _ftext = .;

17 } = 0

18 }

Figure 2.31: chapter2/2/solrex x86 dos.ld

1 # chapter2/2/Makefile

2 #

3 # Author: Wenbo Yang <solrex@gmail.com> <http://solrex.cn>

4 #

5 # This file is part of the source code of book "Write Operating System with

6 # Free Software". The latest version, see <http://share.solrex.cn/WriteOS/>.

7 #

8 # This file is licensed under the GNU General Public License; either

9 # version 3 of the License, or (at your option) any later version.

10

11 CC=gcc

12 LD=ld

13 OBJCOPY=objcopy

14

15 CFLAGS=-c

16 TRIM_FLAGS=-R .pdr -R .comment -R.note -S -O binary

17

18 LDFILE_BOOT=solrex_x86_boot.ld

19 LDFILE_DOS=solrex_x86_dos.ld

20 LDFLAGS_BOOT=-e c -T$(LDFILE_BOOT)

21 LDFLAGS_DOS=-e c -T$(LDFILE_DOS)

22

23 all: boot.img LOADER.BIN

24

25 boot.bin: boot.S

26 $(CC) $(CFLAGS) boot.S

27 $(LD) boot.o -o boot.elf $(LDFLAGS_BOOT)

28 $(OBJCOPY) $(TRIM_FLAGS) boot.elf $@

29

30 LOADER.BIN: loader.S

31 $(CC) $(CFLAGS) loader.S

32 $(LD) loader.o -o loader.elf $(LDFLAGS_DOS)

33 $(OBJCOPY) $(TRIM_FLAGS) loader.elf $@

34

35 boot.img: boot.bin

36 dd if=boot.bin of=boot.img bs=512 count=1

37 dd if=/dev/zero of=boot.img skip=1 seek=1 bs=512 count=2879

38

39 # You must have the authority to do mount, or you must use "su root" or

40 # "sudo" command to do "make copy"

41 copy: boot.img LOADER.BIN

42 mkdir -p /tmp/floppy;\

43 mount -o loop boot.img /tmp/floppy/ -o fat=12;\

44 cp LOADER.BIN /tmp/floppy/;\

45 umount /tmp/floppy/;\

46 rm -rf /tmp/floppy/;

47

48 clean:

49 @rm -f *.o *.elf *.bin *.BIN

50

51 distclean:

52 @rm -f *.img

Figure 2.32: chapter2/2/Makefile

